MATH 333
Lab03Derivatives

Objective
The objective of this project is to illustrate some of the relationships between partial derivatives and tangent lines to surfaces using *Mathematica*.

Narrative
If you have not already done so, read Section 14.3 in the text.

In this project we use the commands **Plot3D** and **ParametricPlot3D**. These commands allows us to plot the graphs of a surface and parametrized space curves in the x and y-directions. Also, we illustrate how we can visualize the partial derivatives as slopes to tangent lines in these directions.

Task
(a) Type the command lines below into *Mathematica*; they produce a plot of the graph of $f(x, y) = -5x/(x^2 + y^2 + 1)$.

```
(* Partial Derivatives *)

f[x_, y_] = -5*x/(x^2 + y^2 + 1);

plot0 = Plot3D[f[x, y], {x, -2, 4}, {y, -2, 4},
  PlotStyle -> Directive[Green, Opacity[.7]]]
```

(b) Continue by typing the commands below into *Mathematica*; they set the value at which we will be drawing tangent lines.

```
a = 2;
b = 2.5;
```

(c) Continue by typing the commands below; they plot the graph of f over a slightly smaller domain, plot the x-curve (in blue) of f through the point $(a, b, f(a, b))$, compute f_x and $f_x(a, b)$, and then draw the tangent line (in red) to the x-curve of f.

```
plot1 = Plot3D[f[x, y], {x, -2, 4}, {y, -2, b},
   PlotStyle -> Directive[Green, Opacity[.8]]];
curve1 = ParametricPlot3D[{t, b, f[t, b]}, {t, -2, 4},
   PlotStyle -> Directive[Blue, Thickness[.01]]];
f1 = Derivative[1, 0][f];
slope1 = f1[a, b];
tanline1 = ParametricPlot3D[{t + a, b, slope1*t + f[a, b]}, {t, -2, 4},
   PlotStyle -> Directive[Red, Thickness[.01]]];
Show[plot1, curve1, tanline1, ViewPoint -> {6, 10, 4},
   AxesLabel -> {x, y, z}, BoxRatios -> {1, 1, 1}]
```

(d) Continue by typing the commands below; they again plot the graph of f over a slighty smaller domain, plot the y-curve (in blue) of f through the point $(a, b, f(a, b))$, compute f_y and $f_y(a, b)$, and then draw the tangent line (in red) to the y-curve of f.

```
plot2 = Plot3D[f[x, y], {x, -2, a}, {y, -2, 4},
   PlotStyle -> Directive[Green, Opacity[.8]]];
curve2 = ParametricPlot3D[{a, t, f[a, t]}, {t, -2, 4},
   PlotStyle -> Directive[Blue, Thickness[.01]]];
f2 = Derivative[0, 1][f];
slope2 = f2[a, b];
tanline2 = ParametricPlot3D[{a, t + b, slope2*t + f[a, b]}, {t, -2, 4},
   PlotStyle -> Directive[Red, Thickness[.01]]];
Show[plot2, curve2, tanline2, ViewPoint -> {10, 6, 2},
   AxesLabel -> {x, y, z}, BoxRatios -> {1, 1, 1}]
```

(e) Finally, type the command line below into *Mathematica*; it plots the original surface, both curves and both tangent lines. Adjust the graphics to get a good view.

```
Show[plot0, curve1, tanline1, curve2, tanline2,
   AxesLabel -> {x, y, z}, BoxRatios -> {1, 1, 1}]
```

At this point, make a hard copy of your typed input and *Mathematica's* responses. Then, ...

(f) By hand, label each x-curve on the graphics you created as "x-curve", the tangent to each x-curve as "tangent to x-curve", each y-curve on the graphics you created as "y-curve", and the tangent to each y-curve as "tangent to y-curve".